
Dreheisen/DreheisenStromregler2.asc
Version 4
SHEET 1 1668 960
WIRE 512 -400 16 -400
WIRE 144 -352 112 -352
WIRE 512 -352 512 -400
WIRE 512 -352 480 -352
WIRE 16 -272 16 -400
WIRE 144 -272 16 -272
WIRE 688 -272 480 -272
WIRE 736 -272 688 -272
WIRE 688 -240 688 -272
WIRE 736 -240 736 -272
WIRE 112 -192 112 -352
WIRE 112 -192 16 -192
WIRE 144 -192 112 -192
WIRE 496 -192 480 -192
WIRE 16 -144 16 -192
WIRE 688 -144 688 -176
WIRE 144 -112 112 -112
WIRE 512 -112 480 -112
WIRE 512 -80 512 -112
WIRE 560 -80 512 -80
WIRE 576 -80 560 -80
WIRE 736 -80 736 -160
WIRE 736 -80 656 -80
WIRE 176 -48 160 -48
WIRE 384 -48 176 -48
WIRE 512 -48 512 -80
WIRE 512 -48 464 -48
WIRE 736 -48 736 -80
WIRE 160 -32 160 -48
WIRE 560 -32 560 -80
WIRE 560 48 560 32
WIRE 736 48 736 32
FLAG 736 48 0
FLAG 688 -144 0
FLAG 16 -64 0
FLAG 160 48 0
FLAG 176 -48 Input
FLAG 560 -192 0
FLAG 112 -112 0
FLAG 560 48 0
SYMBOL ind 720 -256 R0
SYMATTR InstName L_Messwerk
SYMATTR Value 1m
SYMATTR SpiceLine Rser=30
SYMBOL res 720 -64 R0
SYMATTR InstName R1
SYMATTR Value 10
SYMBOL schottky 704 -176 R180
WINDOW 0 24 64 Left 2
WINDOW 3 24 0 Left 2
SYMATTR InstName D1
SYMATTR Value 1N5819
SYMATTR Description Diode
SYMATTR Type diode
SYMBOL voltage 16 -160 R0
WINDOW 123 0 0 Left 2
WINDOW 39 0 0 Left 2
SYMATTR InstName V1
SYMATTR Value 15
SYMBOL voltage 160 -48 R0
WINDOW 123 0 0 Left 2
WINDOW 39 24 124 Left 2
SYMATTR InstName V2
SYMATTR Value PULSE(2.5 0 0 1 1 0)
SYMBOL res 672 -96 R90
WINDOW 0 0 56 VBottom 2
WINDOW 3 32 56 VTop 2
SYMATTR InstName R7
SYMATTR Value 10k
SYMBOL cap 560 -208 R90
WINDOW 0 0 32 VBottom 2
WINDOW 3 32 32 VTop 2
SYMATTR InstName C3
SYMATTR Value 330p
SYMBOL Special\\mc34063 384 -256 R0
SYMATTR InstName U2
SYMBOL res 0 -288 R0
SYMATTR InstName R8
SYMATTR Value 1
SYMBOL res 480 -64 R90
WINDOW 0 0 56 VBottom 2
WINDOW 3 32 56 VTop 2
SYMATTR InstName R2
SYMATTR Value 10k
SYMBOL cap 544 -32 R0
SYMATTR InstName C1
SYMATTR Value 1µ
TEXT 526 -308 Left 2 !.tran 0 2 0 100n startup
TEXT 528 -344 Left 2 !.inc MC34063.lib
TEXT 528 -384 Left 2 !.options Gmin=1e-9  method=Gear




Dreheisen/DreheisenStromregler2.log
Circuit: * C:\Programme\LTC\LTspiceIV\DreheisenStromregler2.asc

Direct Newton iteration failed to find .op point.  (Use ".option noopiter" to skip.)
Starting Gmin stepping
Gmin = 10
Gmin = 1.07374
Gmin = 0.115292
Gmin = 0.0123794
Gmin = 0.00132923
Gmin = 0.000142725
Gmin = 1.5325e-005
Gmin = 1.6455e-006
Gmin = 1.76685e-007
Gmin = 1.89714e-008
Gmin = 2.03704e-009
Gmin = 2.18725e-010
Gmin = 0
Gmin stepping succeeded in finding the operating point.

Heightened Def Con from 6.73601e-007 +++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 +++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 ++++++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 +++++++++++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 ++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 ++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 +++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 +++++++++++++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 +++++++++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 ++++++++++++to 6.73601e-007
Heightened Def Con from 6.73601e-007 to 6.80601e-007

Date: Sat Mar 06 23:32:37 2021
Total elapsed time: 5048.672 seconds.

tnom = 27
temp = 27
method = Gear
totiter = 87936385
traniter = 87935805
tranpoints = 31152681
accept = 26292094
rejected = 4860680
matrix size = 41
fillins = 16
solver = Normal
Matrix Compiler1: 2.17 KB object code size  4.1/3.4/[2.4]
Matrix Compiler2: 3.25 KB object code size  3.1/4.2/[2.1]






Dreheisen/DreheisenStromregler2.net
* C:\Programme\LTC\LTspiceIV\DreheisenStromregler2.asc
L_Messwerk N003 N006 1m Rser=30
R1 N006 0 10
D1 0 N003 1N5819
V1 N002 0 15
V2 Input 0 PULSE(2.5 0 0 1 1 0)
R7 N006 N005 10k
C3 0 N004 330p
XU2 N001 N003 N004 0 N005 N002 N001 N002 MC34063
R8 N001 N002 1
R2 N005 Input 10k
C1 N005 0 1µ
.model D D
.lib C:\Programme\LTC\LTspiceIV\lib\cmp\standard.dio
.tran 0 2 0 100n startup
.inc MC34063.lib
.options Gmin=1e-9  method=Gear
.backanno
.end




Dreheisen/DreheisenStromregler2.op.raw


Dreheisen/DreheisenStromregler2.plt
[Transient Analysis]
{
   Npanes: 1
   {
      traces: 3 {524291,0,"V(n003)"} {524292,0,"V(n005)*1000-1240"} {34603010,1,"I(L_messwerk)"}
      X: (' ',1,0,0.1,1.02120639712065)
      Y[0]: (' ',0,-90,10,60)
      Y[1]: ('m',0,-0.06,0.03,0.39)
      Volts: (' ',0,0,0,-90,10,60)
      Amps: ('m',0,0,0,-0.06,0.03,0.39)
      Log: 0 0 0
      GridStyle: 1
   }
}




Dreheisen/mc34063.asy
Version 4
SymbolType CELL
LINE Normal -97 -80 -82 -80
LINE Normal -97 -48 -97 -80
LINE Normal -82 -48 -97 -48
LINE Normal -82 -48 -82 -48
LINE Normal -82 -48 -82 -48
LINE Normal -49 -64 -66 -64
LINE Normal -112 -30 -57 -30
LINE Normal -112 -54 -112 -30
LINE Normal -97 -54 -112 -54
LINE Normal -170 -71 -97 -71
LINE Normal 8 -64 -10 -64
LINE Normal 8 -56 8 -71
LINE Normal 16 -73 8 -67
LINE Normal 16 -96 16 -73
LINE Normal -176 -96 16 -96
LINE Normal 16 -56 8 -61
LINE Normal 16 -47 16 -56
LINE Normal 28 -47 16 -47
LINE Normal 28 -39 28 -54
LINE Normal 36 -56 28 -50
LINE Normal 36 -39 28 -44
LINE Normal 36 -85 36 -56
LINE Normal 72 -85 36 -85
LINE Normal 36 -29 36 -39
LINE Normal 73 -29 36 -29
LINE Normal -73 -3 -73 -30
LINE Normal -128 -27 -220 -27
LINE Normal -128 -3 -128 -27
LINE Normal -121 127 -121 76
LINE Normal -157 101 -121 127
LINE Normal -121 76 -157 101
LINE Normal -78 93 -121 93
LINE Normal -106 114 -121 114
LINE Normal -170 -71 -170 101
LINE Normal -170 101 -157 101
LINE Normal -106 144 -106 115
LINE Normal 27 144 -106 144
LINE Normal 36 19 -33 19
LINE Normal 36 64 36 19
LINE Normal 56 64 36 64
LINE Normal -129 98 -129 88
LINE Normal -124 93 -134 93
LINE Normal -124 114 -132 114
RECTANGLE Normal 96 160 -240 -112
RECTANGLE Normal -10 -15 -49 -80
RECTANGLE Normal -32 53 -141 -3
RECTANGLE Normal -4 120 -78 71
CIRCLE Normal -56 -34 -49 -26
ARC Normal -97 -80 -66 -48 -82 -48 -82 -80
TEXT -49 -64 Left 0 S
TEXT -10 -64 Right 0 Q
TEXT -47 -30 Left 0 R
TEXT -138 13 Left 0 Ipk
TEXT -42 84 Center 0 1.25V
TEXT -42 106 Center 0 Ref
TEXT -138 36 Left 0 Osc
TEXT -65 16 Left 0 C
TEXT -51 24 Left 0 T
WINDOW 0 -95 -128 Left 0
WINDOW 3 -112 176 Left 0
SYMATTR Value MC34063
SYMATTR Prefix X
SYMATTR Description Switching regulator
PIN 96 -96 RIGHT 8
PINATTR PinName SWC
PINATTR SpiceOrder 1
PIN 96 -16 RIGHT 8
PINATTR PinName SWE
PINATTR SpiceOrder 2
PIN 96 64 RIGHT 8
PINATTR PinName Ct
PINATTR SpiceOrder 3
PIN -240 144 LEFT 8
PINATTR PinName GND
PINATTR SpiceOrder 4
PIN 96 144 RIGHT 8
PINATTR PinName Cinv
PINATTR SpiceOrder 5
PIN -240 64 LEFT 8
PINATTR PinName Vcc
PINATTR SpiceOrder 6
PIN -240 -16 LEFT 8
PINATTR PinName Isns
PINATTR SpiceOrder 7
PIN -240 -96 LEFT 8
PINATTR PinName DRC
PINATTR SpiceOrder 8




Dreheisen/mc34063.lib
* Models adapted to LTSPICE syntax by Helmut Sennewald.  03/14/2004
* Don't remove this copyright notice.
* The used models are property of ONSEMI and/or Intusoft(www.intusoft.com).
* They are from ONsemi's webpage and also from the Intusoft's SPICE (ICAP/4 demo 8.3.10).
* The links are www.onsemi.com and www.intusoft.com .

* All SPICE command lines starting with a comment "*" have been changed 
* to be compatible with LTSPICE. 

* This model is included in the ICAP/4 demo free downloadable software.
* It's very similar to the model in Hymovitz.lib. The vt threshholds
* are different and the model here uses net 90 instead of 0.
.SUBCKT MC34063 swc    swe    ct    90    2    vdd    isns    drc
*               SW-col SW-em  Ct    gnd    cinv vdd    isns    drive col
*DC-DC controller
*B5 5 90 V=~(v(9,90)&v(8,90))
B5 5 90 V=!(v(9,90)&v(8,90))
Q3 ct 22 vdd QN2907
V10 isns 22 DC=.3215
*B6 7 90 V=~(v(4,90)&v(10,90))
B6 7 90 V=!(v(4,90)&v(10,90)) 
R3 5 10 100
R9 13 swe 100
C2 10 90 100p IC=5
R4 2 90 10MEG
C10 2 90 50p
R5 7 8 100
C3 8 90 100p IC=0
S1 srst 90 ct 90 _S2_mod  
R1 srst vdd 10k
Q2 drc 14 13 _Q3_mod 
*B4 6 90 V=v(2,90)  > (v(vref,90) + v(voff,90))  ? 0 : v(vdd,90)
B4 6 90 V=IF(v(2,90)  > (v(vref,90) + v(voff,90)), 0, v(vdd,90) )
*B2 vdd ct I=V(srst,90) > 3 ? 35U : -220U
B2 vdd ct I=IF(V(srst,90) > 3, 35U, -220U )
*B3 9 90 V=v(diff,90) < 1 ? v(vdd,90) : (v(6,90) > 3) ? 0 : v(vdd,90)
B3 9 90 V=IF(v(diff,90) < 1, v(vdd,90), IF( (v(6,90) > 3), 0, v(vdd,90) ) )
*B7 16 90 V=V(vdd,90)-1.5 > 1.25 ? 1.25 : V(vdd,90)-1.25 < 0 ? 0 :V(vdd,90)-1.25
B7 16 90 V=IF( V(vdd,90)-1.5 > 1.25, 1.25, IF( V(vdd,90)-1.25 < 0, 0, V(vdd,90)-1.25 ) )
*V7 16 vref
V7 16 vref 0
R6 vref 90 400
R7 vref vdd 90k
Q1 swc 13 swe _Q4_mod 
R8 diff 90 10k
D1 14 15 DN4148
D2 swe 14 DN4148
V3 drc 15 DC=700m
*B8 swe 14 I=v(5,90) > 2.5 ? 10m : -10m
*** Threshold changed to 0.5, because of different(0,1 versus 0.3,3.5) output level of B-sources.
B8 swe 14 I=IF( v(5,90) > 0.5, 10m, -10m )
V4 voff 90 DC=2m
C5 srst diff 10p
*B1 4 90 V=(v(diff,90) > -1) ? v(vdd,90) : 0
B1 4 90 V=IF( (v(diff,90) > -1), v(vdd,90), 0 )
R10 vdd isns 10k

** discrete models **
.MODEL QN2907 PNP BF=200 BR=6 CJC=19PF CJE=23PF IKF=100E-3
+ IS=1.1E-12 ISE=1.3E-11 MJC=.2 MJE=1.25 NE=1.9 NF=1.21 RC=.6
+ TF=5E-10 TR=34E-9 VAF=50 VJC=.5 VJE=.30 XTB=1.5
.MODEL DN4148 D BV=100V CJO=4PF IS=7E-09 M=.45 N=2 RS=.8
+ TT=6E-09 VJ=.6V
.MODEL _Q4_mod NPN BF=50 RC=.125 RE=.125 TF=10n
.MODEL _Q3_mod NPN BF=50
.MODEL _S2_mod SW VT=1.25 VH=.8

.ENDS
*
*

* This model is part of a ICAP/4 demo circuit from the ONsemi webpage.
* The ground pin must be always at net 0 in the schematic!
.SUBCKT mc34063a 1 2 3 4 5 6 7 8
*exempt 20030227 20395 -30736294
*BY KEHINDE OMOLAYO 2-20-03
*TERMINAL ID
*SWITCH COLLECTOR=1 SWITCH EMITTER=2 TIMING CAPACITOR=3 GND=4
*COMPARATOR INVERTING INPUT=5 VCC=6 IPK SENSE=7 DRIVER COLLECTOR=8
.MODEL DMC34063 D (CJO=2P N=0.05)
.MODEL QSWITCH NPN BF=75 CJC=2P IS=3E-9 RB=1 RC=0.45 RE=0
+ VJC=.75 VJE=.75 VJS=.75
E1 10 0 5 4 1
R1 4 5 10MEG
V1 20 0 PULSE 0 2
E2 11 0 3 4 1
*B1 13 0 V=1M/(ABS((27.475-195M*V(12))+(36.002+244M*V(12))*V(9)-(302.302+651M*V(12))*V(9)^2)+1F)
B1 13 0 V=1M/(ABS((27.475-195M*V(12))+(36.002+244M*V(12))*V(9)-(302.302+651M*V(12))*V(9)**2)+1F)
E3 12 0 6 4 1
E4 9 0 6 7 1
R2 6 7 10MEG
*B2 14 0 V=1M*((-10.765-151M*V(12))+(45.344+864M*V(12))*V(9)-+(35.99+1.378*V(12))*V(9)^2+(8.341+839M*V(12))*V(9)^3)
B2 14 0 V=1M*((-10.765-151M*V(12))+(45.344+864M*V(12))*V(9)-+(35.99+1.378*V(12))*V(9)**2+(8.341+839M*V(12))*V(9)**3)
*B4 15 0 V=V(9)>0.32 ? V(14) : V(13)
B4 15 0 V=IF( V(9)>0.32, V(14), V(13) )
C1 19 0 10P
*B5 16 0 V= V(20)<1 ? 2 :  V(24)>1 ? 2 : V(19)>1 ? 0 : 2
B5 16 0 V= IF( V(20)<1, 2, IF(V(24)>1, 2, IF( V(19)>1, 0, 2 ) ) )
R3 16 17 150
C2 17 0 10P
*B6 18 0 V=V(20)<1 ? 0 : V(24)>1 ? 0 : V(11)<(1.083-1.239*V(29)) ? 2 : V(17)>1 ? 0 : 2
B6 18 0 V= IF( V(20)<1, 0, IF( V(24)>1, 0, IF( V(11)<(1.083-1.239*V(29)), 2, IF(V(17)>1, 0, 2 ) ) ) )
R4 18 19 150
D1 4 3 DMC34063
D2 3 6 DMC34063
C3 11 31 1N
*B7 4 36 I=V(17)>1 ? -(224.4U+2.359U*V(12))*0.77 : V(15)*0.77
B7 4 36 I=IF( V(17)>1, -(224.4U+2.359U*V(12))*0.77, V(15)*0.77 )
C5 23 0 10P
*B9 21 0 V= V(20)<1 ? 2 : V(17)>1 ? 2 : V(26)>1 ? 0 : 2
B9 21 0 V= IF( V(20)<1, 2, IF( V(17)>1, 2, IF( V(26)>1, 0, 2 ) ) )
R5 27 26 150
C6 26 0 10P
*B10 27 0 V=V(20)<1 ? 0 : V(17)>1 ? 0 : V(10)<1.25 ? 2 : V(23)>1 ? 0 : 2
B10 27 0 V=IF( V(20)<1, 0, IF(V(17)>1, 0, IF(V(10)<1.25, 2, IF(V(23)>1, 0, 2 ) ) ) )
R6 21 23 150
*B12 33 0 V=V(11)>(1.148+184.6M*V(29)) ? 2 : 0
B12 33 0 V=IF( V(11)>(1.148+184.6M*V(29)), 2, 0 )
R13 33 24 10K
C8 24 0 10P
Q1 8 30 25 QSWITCH
Q2 1 25 2 QSWITCH
R15 25 2 100
D5 2 30 DMC34063
G1 2 30 26 23 5M
R16 2 4 10MEG
R23 31 28 1M
*V6 28 0 
V6 28 0 0
*V7 36 3 
V7 36 3 0
B13 29 0 V=I(V6)/(I(V7)+866.8M*I(V6))
.ENDS


*==========================================================
* MC34063
* ON Semiconductor
* DC-DC controller
*
* This model was developed for ON Semiconductor by:
* AEI Systems, LLC
* 5777 W. Century Blvd. Suite 876
* Los Angeles, California 90045
* Copyright 2002, all rights reserved.
*
* This model is subject to change without notice.
* Users may not directly or indirectly re-sell or 
* re-distribute this model. This model may not 
* be used, modified, or altered 
* without the consent of ON Semiconductor. 
*
* For more information regarding modeling services,
* model libraries and simulation products, please
* call AEi Systems at (310) 863-8034, or contact 
* AEi by email: info@aeng.com. http://www.AENG.com
*
* Revision: 1.0
*==========================================================
**********
.SUBCKT MC34063p swc    swe    ct    90    2    vdd    isns    drc
*               SW-col SW-em  Ct    gnd    cinv vdd    isns    drive col
*DC-DC controller

* PSpice translation by Christophe Basso, christophe.basso@onsemi.com

EB5 5 90 Value = { IF ( v(9,90)>2.5 & v(8,90)>2.5, 0, 5 ) }
Q3 ct 22 vdd QN2907
V10 isns 22 DC=.3215
EB6 7 90 Value = { IF ( v(4,90)>2.5 & v(10,90)>2.5, 0, 5 ) }
R3 5 10 100
R9 13 swe 100
C2 10 90 100p IC=5
R4 2 90 10MEG
C10 2 90 50p
R5 7 8 100
C3 8 90 100p IC=0
XTOF1 srst 90 ct 90 SWhyste params: VT=1.25 VH=.8 
R1 srst vdd 10k
Q2 drc 14 13 _Q3_mod 
EB4 6 90 Value = { IF ( v(2,90)  > (v(vref,90) + v(voff,90)), 0, v(vdd,90) ) }
GB2 vdd ct Value = { IF ( V(srst,90) > 3, 35U, -220U ) }
EB3 9 90 Value = { IF ( v(diff,90) < 1, v(vdd,90), IF ( (v(6,90) > 3), 0, v(vdd,90) ) ) }
EB7 16 90 Value = { IF ( V(vdd,90)-1.5 > 1.25, 1.25, IF ( V(vdd,90)-1.25 < 0, 0, V(vdd,90)-1.25 ) ) }
V7 16 vref
R6 vref 90 400
R7 vref vdd 90k
Q1 swc 13 swe _Q4_mod 
R8 diff 90 10k
D1 14 15 DN4148
D2 swe 14 DN4148
V3 drc 15 DC=700m
GB8 swe 14 Value = { IF ( v(5,90) > 2.5, 10m, -10m ) }
V4 voff 90 DC=2m
C5 srst diff 10p
EB1 4 90 Value = { IF ( (v(diff,90) > -1), v(vdd,90), 0 ) }
R10 vdd isns 10k
** discrete models **
.MODEL QN2907 PNP BF=200 BR=6 CJC=19PF CJE=23PF IKF=100E-3
+ IS=1.1E-12 ISE=1.3E-11 MJC=.2 MJE=1.25 NE=1.9 NF=1.21 RC=.6
+ TF=5E-10 TR=34E-9 VAF=50 VJC=.5 VJE=.30 XTB=1.5
.MODEL DN4148 D BV=100V CJO=4PF IS=7E-09 M=.45 N=2 RS=.8
+ TT=6E-09 VJ=.6V
.MODEL _Q4_mod NPN BF=50 RC=.125 RE=.125 TF=10n
.MODEL _Q3_mod NPN BF=50
.ENDS
**********
* PSpice hysteresis switch made by Chris Basso
.subckt SWhyste NodeMinus NodePlus Plus Minus PARAMS: RON=1 ROFF=1MEG VT=5 VH=2
S5 NodePlus NodeMinus 8 0 smoothSW
EBcrtl 8 0 Value = { IF ( V(plus)-V(minus) > V(ref), 1, 0 ) }
EBref ref1 0 Value = { IF ( V(8) > 0.5, {VT-VH}, {VT+VH} ) }
Rdel ref1 ref 100
Cdel ref 0 100p  IC={VT+VH}
Rconv1 8 0 10Meg
Rconv2 plus 0 10Meg
Rconv3 minus 0 10Meg
.model smoothSW VSWITCH (RON={RON} ROFF={ROFF} VON=1 VOFF=0)
.ends SWhyste

* Hymovitz.lib
* This model is very similar to the model in the 
*==========================================================
* MC34063
* ON SEMICONDUTOR
* 1.5 A, Step-Up/Down/Inverting Switching Regulator
*
* This model was developed for On Semiconductor by:
* AEI Systems, LLC
* 5777 W. Century Blvd. Suite 876
* Los Angeles, California 90045
* Copyright 2002, all rights reserved.
*
* This model is subject to change without notice.
* Users may not directly or indirectly re-sell or 
* re-distribute this model. This model may not 
* be used, modified, or altered 
* without the consent of On Semiconductor. 
*
* For more information regarding modeling services,
* model libraries and simulation products, please
* call AEi Systems at (310) 863-8034, or contact 
* AEi by email: info@aeng.com. http://www.AENG.com
*
* Revision: 2/18/02, version 1.1
*==========================================================
**********
*SRC=MC33063;MC33063;Regulators;On Semiconductor;1.5A
*SYM=MC34063
.SUBCKT MC33063 swc    swe    ct    90    2    vdd    isns    drc
*               SW-col SW-em  Ct    gnd    cinv vdd    isns    drive col
*DC-DC controller
*B5 5 0 V=~(v(9)&v(8))
B5 5 90 V=!(v(9,90)&v(8,90))
Q1 ct isns vdd QN2907
.MODEL QN2907 PNP BF=200 BR=6 CJC=19PF CJE=23PF IKF=100E-3
+ IS=1.1E-12 ISE=1.3E-11 MJC=.2 MJE=1.25 NE=1.9 NF=1.21 RC=.6
+ TF=5E-10 TR=34E-9 VAF=50 VJC=.5 VJE=.85 XTB=1.5
*B6 7 0 V=~(v(4)&v(10))
B6 7 90 V=!(v(4,90)&v(10,90)) 
R3 5 10 100
R9 13 swe 100
*C2 10 0 100p IC=5
C2 10 90 100p IC=5
R4 2 90 10MEG
R5 7 8 100
*C3 8 0 100p IC=0
C3 8 90 100p IC=0
S1 srst 90 ct 90 _S2_mod  
.MODEL _S2_mod SW VT=1.75 VH=1.25
R1 srst vdd 10k
Q2 drc 14 13 _Q3_mod 
.MODEL _Q3_mod NPN BF=50
*B4 6 0 V=v(2,90)  > (v(vref,90) + v(voff,90))  ? 0 : v(vdd)
B4 6 90 V=IF( v(2,90)  > (v(vref,90) + v(voff,90)), 0, v(vdd,90) )
*B2 vdd ct I=V(srst,90) > 3 ? 35U : -220U
B2 vdd ct I=IF( V(srst,90) > 3, 35U, -220U )
*B3 9 0 V=(v(6,90) > 3) ? v(diff,90) > 1 ? 0 : v(vdd)
B3 9 90 V=IF( (IF((v(6,90) > 3),1,0) & IF((v(diff,90) > 1),1,0)), 0, v(vdd,90) ) 
*B7 16 90 V=V(vdd,90)-1.5 > 1.25 ? 1.25 : V(vdd,90)-1.25 < 0 ? 0 :V(vdd,90)-1.25
B7 16 90 V=IF( V(vdd,90)-1.5 > 1.25, 1.25, IF( V(vdd,90)-1.25 < 0, 0, V(vdd,90)-1.25 ) )
*V7 16 vref
V7 16 vref 0
R6 vref 90 400
R7 vref vdd 90k
Q1x swc 13 swe _Q4_mod 
.MODEL _Q4_mod NPN BF=50 RC=.25 RE=.25 TF=0
R8 diff 90 10k
D1 14 15 DN4148
.MODEL DN4148 D BV=100V CJO=4PF IS=7E-09 M=.45 N=2 RS=.8
+ TT=6E-09 VJ=.6V
D2 swe 14 DN4148
V3 drc 15 DC=700m
*B8 swe 14 I=v(5) > 2.5 ? 1m : -1m
*** Threshold changed to 0.5, because of different(0,1 versus 0.3,3.5) output level of B-sources.
B8 swe 14 I=IF( v(5,90) > 0.5, 1m, -1m )
V4 voff 90 DC=2m
C5 srst diff 10p
*B1 4 90 V=(v(6,90) > 3) ? (v(diff,90) > -1) ? v(vdd) : 0
B1 4 90 V=IF( (IF((v(6,90) > 3),1,0) & IF((v(diff,90) > -1),1,0)), v(vdd,90),0 ) 
R10 vdd isns 10k
.ENDS
**********




***** Example models *****
.MODEL DN5819 D BV=5.33E+01 CJO=1.44E-10 EG=0.69
+ IBV=6.00E-04 IS=1.65E-05 M=.671 N=1.41 RS=4.47E-02
+ TT=7.20E-11 VJ=1.45 XTI=2
*****



